# $Id: PKGBUILD 67738 2012-03-14 10:05:54Z giovanni $
# Maintainer: Jaroslav Lichtblau <dragonlord@aur.archlinux.org>
# Contributor: Allan McRae <allan@archlinux.org>
# Contributor: whatah <whatah@gmail.com>
# Contributor: Andrew Conkling <andrewski@fr.st>

pkgname=pympd
pkgver=0.08.1
pkgrel=3
pkgdesc="A Rhythmbox-like PyGTK+ client for Music Player Daemon"
arch=('i686' 'x86_64')
url="http://pympd.sourceforge.net/"
license=('GPL')
depends=('libglade' 'pygtk')
makedepends=('pkgconfig')
options=('!emptydirs')
source=(http://pympd.sourceforge.net/files/$pkgname-$pkgver.tar.gz)
md5sums=('388c2882ffbacffb27b491441a2b2a23')

package() {
  cd ${srcdir}/$pkgname-$pkgver

  # python2 fix
  sed -i 's_which python_which python2_' Makefile
  sed -i 's_python -c_python2 -c_' src/modules/tray/Makefile
  for file in $(find . -name '*.py' -print); do
    sed -i 's_^#!.*/usr/bin/python_#!/usr/bin/python2_' $file
    sed -i 's_^#!.*/usr/bin/env.*python_#!/usr/bin/env python2_' $file
  done

  make PREFIX=/usr DESTDIR=${pkgdir} install
}
